Exercise 5 – Honeycomb buckling



67

TITLE:
Honeycomb: 4-point bending of honeycomb structure for train carriages.

PROBLEM:
This problem has been taken from the paper listed below in the references and reproduces an experiment of 4-point bending of an extruded aluminum panel used in railway carriages. Loading is very slow (nominally static) in the experiment. In the numerical simulation, we use a somewhat faster imposed displacement for efficiency of the explicit calculation, but still in relatively slow regime.

MESH:
The model is 3D and uses 3712 elements Q4GS for the structure. A localized "defect" is introduced near the centre of the panel (see mesh generation file) to induce the right startup of the dynamic buckling mode.

MATERIALS:
An elasto-plastic VM23 material is used to simulate the aluminum. A failure criterion is introduced.

BOUNDARY CONDITIONS:
The panel is held fixed along two lines, and an imposed displacement is applied along other two lines (4-point bending). This induces some localized buckling of the panel.

LOADING:
The applied "load" is rather an applied displacement. The resulting reaction force (as computed by the code) is monitored and compared with the experimentally measured value.

CALCULATION:

The calculation is performed up to 100 ms. At the final time, the panel has undergone large motions and some localized buckling in its central part.

RESULTS:
Despite the use of a rather coarse mesh (much coarser than the meshes used in the calculations contained in the cited paper), results compare well with the experiment (see cited paper) both as concerns the buckling mode and the measured force. The experimental load curves are given in Figure 10 of the mentioned paper, while the obtained final shape of the panel is shown in Figure 25 of the paper.

POST-TREATMENT
An animation of the computed results from this calculation is available on the EUROPLEXUS Consortium Web site.

REFERENCES:
This problem is detailed in:

1) L. Zheng, D. Petry, T. Wierzbicki, H. Rapp: "Fracture prediction in 4-point bending of an extruded aluminium panel". Thin-Walled Structures 43 (2005), 565-590.

2) M. Larcher: “Simulations of a Metro Carriage Exposed to an Internal Detonation, PUBSY N. JRC50327, 2009.

Numerical Solution

HONEYCOMB

The mesh generation file (K2000):

```
OPTI echo 1;
OPTI dime 3 elem qua4;
den=0.025;
DENS den;
t1 = 0.05;
t2 = 0.145;

p1 = 0 0 0;
p2 = (t2/2.0) 0 0;
p3 = t2 0 0;
p4 = 0 t1 0;
p5 = t2 t1 0;
p6 = (t2*2.0) 0 0;
p7 = (t2*3.0) 0 0;

l1 = p1 p2 d p3 d;
l2 = p4 p5 d;
l3 = p4 p2 d p5 d;
l4 = p1 p4 d;
l5 = l4 plus p7;

l1 = l1 et (l1 plus p3) et (l1 plus p6);
l2 = l2 et (l2 plus p3) et (l2 plus p6);
l3 = l3 et (l3 plus p3) et (l3 plus p6);

pro = l1 et l2 et l3 et l4 et l5;

p9 = 0 0 1.45;

vol = pro tran 58 p9;

lb1 = l1 plus (0 0 0.125);
lb2 = l1 plus (0 0 1.325);
lb3 = 0 0 0.125;
lb4 = (t2*3.0) 0 0.125;
ll1 = l1 plus (0 0 0.525);
ll2 = l1 plus (0 0 0.925);

kod1 = faux;
REPE I0 (NBEL (vol ELEM QUA4));
  xx yy zz = coor (bary (vol ELEM QUA4 &I0));
  SI ( yy ega 0.000 0.0001);
     si (kod1);airb=airb et (vol ELEM QUA4 &I0);
     sinon; airb = vol ELEM QUA4 &I0;kod1 = vrai;
     fins;
  FINS;
FIN I0;

mesh = lb1 et lb2 et lb3 et ll1 et ll2 et vol et airb et lb4;
elim mesh 0.00001;

pcent = bary airb;

pdis1 = airb poin 'PROC' pcent;
edis1 = vol ELEM CONTENANT pdis1;
pdis2 = pdis1 plus p4;
edis2 = vol ELEM CONTENANT pdis2;

depl edis2 plus (0 0.0001 0);

TASS mesh;
OPTI sauv form 'vl_jrc_honeycomb.msh';
sauv form mesh;
*
opti trac psc ftra 'vl_jrc_honeycomb_mesh.ps';
trac cach mesh;
fin;
```

The EUROPLEXUS input file is:

```
honey_comb - quasistatisch
ECHO
!VERI
!CONV WIN
CAST modell
TRID LAGR FAIL 1.0
OPTI TOLC 1e-1
$
DIME
  PT6L        3422     ZONE  2
  Q4GS        3712
  ECRO      892040
  TABL 1 2
  DEPL 38
TERM
$
GEOM
  Q4GS vol
TERM
$
COMP EPAI 0.003 LECT vol TERM
     COUL vert  LECT vol TERM
MATE VM23 RO 2700 YOUN 7E10 NU 0.3
          ELAS 200E6
          FAIL VMIS LIMI 245e6
          TRAC 2 200e6    2.85e-3
                 250e6    0.23
          LECT vol TERM
LINK COUP BLOQ 123 LECT lb3 TERM
          BLOQ  23 LECT lb4 TERM
          BLOQ   2 LECT lb1 lb2 TERM
CHAR 1 FACT 2 DEPL 2 0.08 LECT ll1 ll2 TERM
       TABL 2 0 0 1.1e-1 -1.1
REGI 'lb1' TOUT POIN LECT lb1 TERM
     'lb2' TOUT POIN LECT lb2 TERM
     'll1' TOUT POIN LECT ll1 TERM
     'll2' TOUT POIN LECT ll2 TERM
ECRI FICH ALIC tfreq 4e-3
     FICH ALIC TEMP tfreq 1e-4
               POIN LECT pdis1 TERM
               ELEM LECT edis1 ll1 ll2 lb1 lb2 TERM
     FICH FORM PVTK tFREQ 4e-3 VARI ECRO DEPL FAIL CONT
               GROU 1 OBJE LECT vol TERM
$
OPTI NOTE LOG 1
$
CALC TINI 0 TEND 1e-1
*================================================================
SUIT
Post-treatment (time curves from alice temps file)
ECHO
*
RESU ALIC TEMP GARD PSCR
SORT GRAP
AXTE 1.0 'Time [s]'
COUR 1 'resu1' RESU COMP 2 REGI 1
COUR 2 'resu2' RESU COMP 2 REGI 2
COUR 3 'resu3' RESU COMP 2 REGI 3
COUR 4 'resu4' RESU COMP 2 REGI 4
trac 1 2 3 4 axes 1.0 'FORCE [N]'
COUR 10 'dz_pdis' DEPL COMP 2 NOEU LECT pdis1 TERM
trac 10 axes 1.0 'DISPL. [M]'
COUR 5 ADD 1 2
COUR 11 MULC 10 -1.0
TRAC 5 text axes 1.0 'FORCE [N]' XAXE 11 1.0 'DISPL. [M]'
*
QUAL DEPL COMP 2 LECT pdis1 TERM REFE  -1.09634E-01 TOLE 2.E-2
*================================================================
FIN
```

Some results: specimen final shape.





Buckling mode:





Comparison experiment/calculation: reaction forces



Final shape: